INTERNAL REPORTS IN

SIMULATION, OPTIMIZATION AND CONTROL

No. SOC-197

STEADY STATE SOLUTION OF THE MATRIX RICCATI EQUATION

H.L. Abdel-Malek

April 1978

FACULTY OF ENGINEERING
McMASTER UNIVERSITY
HAMILTON, ONTARIO, CANADA

STEADY STATE SOLUTION

OF

THE MATRIX RICCATI EQUATION

H.L. Abdel-Malek

Abstract

A brief review of methods proposed for solving the matrix Riccati equation is given. A FORTRAN listing of a computer program based on Kleinman iterative technique is included. It is shown how to formulate the system of simultaneous linear equations to be solved in the Kleinman method. Some illustrative examples are also given.

This work was supported by the Department of Electrical Engineering, McMaster University, Hamilton, Canada, and by a National Research Council of Canada Postdoctoral Fellowship to the author.

The author is with the Group on Simulation, Optimization and Control and the Department of Electrical Engineering, McMaster University, Hamilton, Canada, L8S 4L7.

I. INTRODUCTION

Consider the nth-order linear system given by

$$\dot{\underset{\sim}{x}} = \underset{\sim}{A}\,\underset{\sim}{x} + \underset{\sim}{B}\,\underset{\sim}{u} \,. \tag{1}$$

The linear time invariant regulator problem can be stated as follows: find the optimal control $\underset{\sim}{u}^*$ which minimizes the cost function (performance index)

$$J = \frac{1}{2}\int_0^\infty (\underset{\sim}{x}^T \underset{\sim}{Q}\,\underset{\sim}{x} + \underset{\sim}{u}^T \underset{\sim}{R}\,\underset{\sim}{u})\,dt \,, \tag{2}$$

where the matrices $\underset{\sim}{Q}$ and $\underset{\sim}{R}$ are both symmetric, and non-negative and positive definite, respectively.

The optimal linear control $\underset{\sim}{u}^*$ which minimizes the cost function J is given by [1],

$$\underset{\sim}{u}^* = -\underset{\sim}{R}^{-1}\,\underset{\sim}{B}^T\,\underset{\sim}{\bar{P}}\,\underset{\sim}{x} \,, \tag{3}$$

where $\underset{\sim}{\bar{P}}$ is the constant n x n matrix which is positive definite and is the solution of the algebraic quadratic matrix equation

$$\underset{\sim}{\bar{P}}\,\underset{\sim}{A} + \underset{\sim}{A}^T\,\underset{\sim}{\bar{P}} - \underset{\sim}{\bar{P}}\,\underset{\sim}{B}\,\underset{\sim}{R}^{-1}\,\underset{\sim}{B}^T\,\underset{\sim}{\bar{P}} + \underset{\sim}{Q} = \underset{\sim}{0} \,. \tag{4}$$

The existence of the steady state (and constant) solution $\underset{\sim}{\bar{P}}$ is guaranteed by the controllability condition of the system $[\underset{\sim}{A},\underset{\sim}{B}]$. The observability condition guarantees the positive definiteness of $\underset{\sim}{\bar{P}}$.

The minimum cost for steering any initial state $\underset{\sim}{x}(0)$ to the origin is given by

$$J^* = \frac{1}{2}\,\underset{\sim}{x}^T(0)\,\underset{\sim}{\bar{P}}\,\underset{\sim}{x}(0) \,, \tag{5}$$

There are several different approaches to obtain the solution of equation (4).

(A) A procedure described by Potter in [2], which requires the computation of the eigenvalues and eigenvectors of the matrix $\underset{\sim}{M}$ given by

$$\underset{\sim}{M} = \left[\begin{array}{c|c} \underset{\sim}{A} & -\underset{\sim}{B}\,\underset{\sim}{R}^{-1}\,\underset{\sim}{B}^T \\ \hline -\underset{\sim}{Q} & -\underset{\sim}{A}^T \end{array}\right] . \quad (6)$$

This matrix has the property that, if λ is an eigenvalue of $\underset{\sim}{M}$ then $-\lambda$ is so. Let the matrix of eigenvectors v be given as

$$\underset{\sim}{V} = \left[\begin{array}{c|c} \underset{\sim}{V}_{11} & \underset{\sim}{V}_{12} \\ \hline \underset{\sim}{V}_{21} & \underset{\sim}{V}_{22} \end{array}\right] \quad (7)$$

and is constructed such that,

$$\underset{\sim}{V}^{-1}\,\underset{\sim}{M}\,\underset{\sim}{V} = \left[\begin{array}{c|c} -\underset{\sim}{\Lambda} & \underset{\sim}{0} \\ \hline \underset{\sim}{0} & \underset{\sim}{\Lambda} \end{array}\right] , \quad (8)$$

where $\underset{\sim}{\Lambda}$ is a Jordan block form of the eigenvalues with positive real parts. Then the desired matrix $\underset{\sim}{\overline{P}}$ is given by

$$\underset{\sim}{\overline{P}} = \underset{\sim}{V}_{21}\,\underset{\sim}{V}_{11}^{-1} . \quad (9)$$

(B) The following procedure was described by Bucy and Joseph [3]. It is based on computing only the eigenvalues of the matrix $\underset{\sim}{M}$. Then a matrix $\underset{\sim}{p}(\underset{\sim}{M})$ is constructed in the following manner,

$$\underset{\sim}{p}(\underset{\sim}{M}) = \underset{\sim}{M}^q + a_1\,\underset{\sim}{M}^{q-1} + a_2\,\underset{\sim}{M}^{q-2} + \dots + a_q\,\underset{\sim}{I}_{2n} , \quad (10)$$

where a_1, a_2, ..., a_q and q are obtained by constructing the polynomial p(s) whose zeros consists of the left half plane (negative real part) eigenvalues of $\underset{\sim}{M}$.

$$p(s) = s^q + a_1 s^{q-1} + a_2 s^{q-2} + \ldots + a_q \ . \tag{11}$$

Then $\underset{\sim}{\overline{P}}$ is uniquely defined by solving the system of linear equations given by

$$\underset{\sim}{p}(\underset{\sim}{M}) \begin{bmatrix} \underset{\sim}{I}_n \\ \\ \underset{\sim}{\overline{P}} \end{bmatrix} = \underset{\sim}{0} \ . \tag{12}$$

The two previous methods require eigenvalue computation which may pose some problems for large-order systems.

(C) An iterative procedure was presented by Kleinman in [4]. This procedure was applied and a computer program is available. Some illustrative examples are also presented (see the Appendix).

A formulation of the system of linear equations for solving the matrix equation

$$\underset{\sim}{A}^T \underset{\sim}{P} + \underset{\sim}{P}\underset{\sim}{A} = \underset{\sim}{W} \tag{13}$$

is presented, it contains n(n+1)/2 equations for the case of symmetric $\underset{\sim}{W}$ and hence $\underset{\sim}{P}$ is also symmetric.

II. THE KLEINMAN ITERATIVE TECHNIQUE FOR SOLVING THE RICCATI EQUATION

The algorithm [4] can be described by the following steps.

Step (1) Select an mxn matrix $\underset{\sim}{L}_0$, where m is the number of inputs and n is the order of the system, such that

$$\underset{\sim}{A}_0 = \underset{\sim}{A} - \underset{\sim}{B}\,\underset{\sim}{L}_0$$

has all its eigenvalues with negative real parts. If all the eigenvalues of $\underset{\sim}{A}$ are with negative real parts $\underset{\sim}{L}_0 = \underset{\sim}{0}$ is an immediate selection. Such an $\underset{\sim}{L}_0$ always exists since the system is completely controllable [5].

Step (2) Solve the system of linear equations given by

$$\underset{\sim}{A}_k^T \underset{\sim}{P}_k + \underset{\sim}{P}_k \underset{\sim}{A}_k + \underset{\sim}{Q} + \underset{\sim}{L}_k^T \underset{\sim}{R}\, \underset{\sim}{L}_k = \underset{\sim}{0}, \quad k = 0, 1, 2, \ldots \tag{14}$$

for $\underset{\sim}{P}_k$. In general, since $\underset{\sim}{P}_k$ is symmetric, this is a system of $n(n+1)/2$ equations.

Step (3) Find the feedback matrix

$$\underset{\sim}{L}_k = \underset{\sim}{R}^{-1} \underset{\sim}{B}^T \underset{\sim}{P}_k \; . \tag{15}$$

Step (4) Calculate,

$$\underset{\sim}{A}_k = \underset{\sim}{A} - \underset{\sim}{B}\,\underset{\sim}{L}_k \tag{16}$$

and go back to Step (2).

It is proved that $\lim_{k\to\infty} \underset{\sim}{P}_k = \overline{\underset{\sim}{P}}$ and that the convergence of the positive definite sequence $\underset{\sim}{P}_1, \underset{\sim}{P}_2, \ldots$ to $\overline{\underset{\sim}{P}}$ is quadratic at the limit [4].

The stopping criterion used in the program is

$$||\underset{\sim}{P}_{k+1}||_\infty - ||\underset{\sim}{P}_k||_\infty < \varepsilon \; , \tag{17}$$

where ε is a prescribed small positive number.

III. FORMULATION OF THE SYSTEM OF LINEAR EQUATIONS

In order to solve the system of simultaneous linear equations (14), the formulation of the matrix of coefficents $\underset{\sim}{H}$ is necessary. The system of linear equations has the following form

$$\underset{\sim}{P}\,\underset{\sim}{A} + \underset{\sim}{A}^T\underset{\sim}{P} = \underset{\sim}{W} \,. \tag{18}$$

Since $\underset{\sim}{W}$ and $\underset{\sim}{P}$ are symmetric, the system of linear equations was formulated to solve for the lower triangle of $\underset{\sim}{P}$ arranged in a columnwise manner as shown in (19), given on the next page.

ACKNOWLEDGEMENT

The author wishes to thank Dr. N.K. Sinha for making this work possible.

REFERENCES

[1] H. Kwakernaak and R. Sivan, Linear Optimal Control Systems. John Wiley: New York, 1972.

[2] J.E. Potter, "Matrix quadratic solutions", SIAM J. Applied Math., vol. 14, No. 3, May 1966, pp. 496-501.

[3] R.S. Bucy and P.D. Joseph. Filtering for Stochastic Processes with Applications to Guidance. Interscience Publishers: New York, 1968.

[4] D.L. Kleinman, "On iterative technique for Riccati equation computation", IEEE Trans. Auto. Control, vol. AC-13, February 1968, pp. 114-115.

[5] W.M. Wonham, "On pole assignement in multi-input controllable linear systems", IEEE Trans. Auto. Control, vol. AC-12, December 1967, pp. 660-665.

$$
\begin{array}{c}
\\ \\ \\ n \\ \\ \\ \\ \\ \\ \\ n-1 \\ \\ \\ \\ \\ \\ \\ n-2 \\ \\ \\ \\ 1
\end{array}
\left[\begin{array}{ccccccc|cccccc|cccc|c}
2a_{11} & 2a_{21} & 2a_{31} & & & \cdots & 2a_{n1} & 0 & 0 & 0 & & \cdots & 0 & 0 & 0 & \cdots & 0 & 0 \\
a_{12} & a_{22}+a_{11} & a_{32} & & & \cdots & a_{n2} & a_{21} & a_{31} & a_{41} & & \cdots & a_{n1} & 0 & 0 & \cdots & 0 & 0 \\
a_{13} & a_{23} & a_{33}+a_{11} & & & \cdots & a_{n3} & 0 & a_{21} & 0 & & \cdots & 0 & a_{31} & a_{41} & \cdots & a_{n1} & 0 \\
\vdots & \vdots & \vdots & & & & \vdots & 0 & 0 & a_{21} & & & 0 & 0 & a_{31} & \cdots & a_{n1} & 0 \\
 & & & & & & & \vdots & \vdots & \vdots & & & & \vdots & \vdots & & & \vdots \\
a_{1n} & a_{2n} & a_{3n} & & & \cdots & a_{nn}+a_{11} & 0 & 0 & 0 & & \cdots & a_{21} & 0 & 0 & \cdots & a_{31} & a_{n1} \\
\hline
0 & 2a_{12} & 0 & 0 & 0 & \cdots & 0 & 2a_{22} & 2a_{32} & 2a_{42} & & \cdots & 2a_{n2} & 0 & 0 & \cdots & 0 & 0 \\
0 & a_{13} & a_{12} & 0 & 0 & \cdots & 0 & a_{23} & a_{33}+a_{22} & a_{43} & & \cdots & a_{n3} & a_{32} & a_{42} & \cdots & a_{n2} & 0 \\
0 & a_{14} & 0 & a_{12} & 0 & \cdots & 0 & a_{24} & a_{34} & a_{44}+a_{22} & & \cdots & a_{n4} & 0 & a_{32} & \cdots & a_{n2} & 0 \\
0 & a_{15} & 0 & 0 & a_{12} & \cdots & 0 & \vdots & \vdots & \vdots & & & \vdots & \vdots & \vdots & & \vdots & \vdots \\
\vdots & \vdots & \vdots & \vdots & \vdots & & \vdots & & & & & & & & & & & \\
0 & a_{1n} & 0 & 0 & 0 & \cdots & a_{12} & a_{2n} & a_{3n} & a_{4n} & & \cdots & a_{nn}+a_{22} & 0 & 0 & \cdots & a_{32} & a_{n2} \\
\hline
0 & 0 & 2a_{13} & 0 & 0 & \cdots & 0 & 0 & 2a_{23} & 0 & 0 & \cdots & 0 & 2a_{33} & 2a_{43} & \cdots & 2a_{n3} & 0 \\
0 & 0 & a_{14} & a_{13} & 0 & \cdots & 0 & 0 & a_{24} & a_{23} & 0 & \cdots & 0 & a_{34} & a_{44}+a_{33} & \cdots & a_{n4} & 0 \\
0 & 0 & a_{15} & 0 & a_{13} & \cdots & 0 & 0 & a_{25} & 0 & a_{23} & \cdots & 0 & a_{35} & a_{45} & \cdots & a_{n5} & \vdots \\
\vdots & \vdots & \vdots & \vdots & \vdots & & \vdots & \vdots & \vdots & \vdots & \vdots & & \vdots & \vdots & \vdots & & \vdots & \vdots \\
0 & 0 & a_{1n} & 0 & 0 & \cdots & a_{13} & 0 & a_{2n} & 0 & 0 & \cdots & a_{23} & a_{3n} & a_{4n} & \cdots & a_{nn}+a_{33} & a_{n3} \\
\hline
 & & & & & & & & & & & & & & & & & \\
\hline
0 & 0 & 0 & 0 & 0 & \cdots & 2a_{1n} & 0 & 0 & 0 & & \cdots & 2a_{2n} & 0 & 0 & \cdots & 2a_{3n} & 2a_{nn}
\end{array}\right]
\left[\begin{array}{c}
p_{11} \\ p_{12} \\ p_{13} \\ \vdots \\ \\ p_{1n} \\ \hline p_{22} \\ p_{32} \\ p_{42} \\ \vdots \\ \\ p_{n2} \\ \hline p_{33} \\ p_{43} \\ \vdots \\ \\ p_{n3} \\ \hline \\ \hline p_{nn}
\end{array}\right]
=
\left[\begin{array}{c}
w_{11} \\ w_{12} \\ w_{13} \\ \vdots \\ \\ w_{1n} \\ \hline w_{22} \\ w_{32} \\ w_{42} \\ \vdots \\ \\ w_{n2} \\ \hline w_{33} \\ w_{43} \\ \vdots \\ \\ w_{n3} \\ \hline \\ \hline w_{nn}
\end{array}\right]
\qquad (19)
$$

$$
\qquad\qquad n \qquad\qquad\qquad\qquad n-1 \qquad\qquad\qquad\qquad n-2 \qquad\qquad 1
$$

APPENDIX

THE COMPUTER PROGRAM

MRICKL is a package of subroutines for solving the Matrix Raccati equaiton in the steady state using Kleinman technique.

<u>Argument List</u>

SUBROUTINE MRICKL (A, B, Q, R, SL, P, PP, Z, AA, RINV, L, N, M, NU, MU, INP, IPT, EPS, MAX, KR, H).

The arguments are as follows,

N An integer to be set to the order of the system.

M An integer to be set to the number of inputs.

NU N(N+1)/2.

MU M(M+1)/2.

A An NxN matrix of the system.

B An NxM matrix of the system.

Q The symmetric matrix used fro weighting the states. It is to be stored in a vector form of dimension NU and arranged columnwise in upper triangular form as shown.

$$\begin{bmatrix} 1 & 2 & 4 & 7 \\ 2 & 3 & 5 & 8 \\ 4 & 5 & 6 & 9 \\ 7 & 8 & 9 & 10 \end{bmatrix}$$

Storage of symmetric matrices in vector form.

R The positive definite symmetric matrix for weighting the inputs. It is to be stored in a vector form of dimension MU and arranged columnwise in upper triangle form.

On return R contains the upper triangle factorization of R ($R=U^TU$).

SL An MxN matrix which contains the feedback matrix at each iteration.

P The solution of the matrix Riccati equation stored in columnwise form having of dimension NU.

PP A vector of dimension NU in which the previous value of P is stored.

Z A working area of dimension MxN.

AA A working area of dimension NxN, for storing $A_k = A-B\ (SL)_k$.

RINV A vector of dimension MU in which the upper triangle of the inverse of R is stored in a columnwise manner.

KR An integer column vector of dimension NU used as working space.

H A matrix of dimension NUxNU in which the coefficients of the system of linear equations are stored.

MAX An integer to be set to the maximum number of iterations allowed.

EPS A real number to be set to the test quantity used for checking the accuracy of inverting R and in solving the system of linear equation. If this accuracy is not satisfied a warning will be given. Also, EPS is used for stopping the iteration loop.

INP An integer to be set to 0 if input data are not to be printed. Otherwise, set to 1.

IPT An integer controlling printing of intermediate output. Printing occurs every |IPT| iterations and also on exit except when IPT is set to zero, in this case intermediate output is suppressed.

L An integer to be set to zero if $(SL)_0 = 0$, otherwise $(SL)_0$ must be given.

All symmetric matrices are printed in lower triangular form. All matrices printing format are arranged for a maximum number of columns equal 10. If better printing is required for higher orders, format modification is required.

The total memory required = $(N^2/4)(N+1)^2 + 2N(2N+1) + M(3N+M+1)$.

EXAMPLES

Example 1

A system of order N=3 is given and it is required to find the optimal feedback matrix. It is necessary to calculate the feedback matrix $\underset{\sim}{SL}$ according to the final value of $\overline{\underset{\sim}{P}}$. Thus, the following two cards were added.

```
CALL NEWSL(RINV, B, P, SL, N, M) ,
```
$\underset{\sim}{SL} = \underset{\sim}{R}^{-1} \underset{\sim}{B}^{T} \underset{\sim}{P}$.

```
CALL MPRINT(SL, M, N) ,
```
for printing.

The main program, input data and final solution are given. It is to be noted that the $\underset{\sim}{A}$ matrix has all eigenvalues with negative real parts.

```
      DIMENSION A(3,3),B(3,2),Q(6),R(3),SL(2,3),P(6),PP(6),Z(2,3),
     +          AA(3,3),RINV(3),KR(6),H(6,6)
C
      DATA A/-0.6,0.0,4.0,0.0,-4.0,-1.6,0.45,0.8,-0.8/
      DATA B/-0.45,0.0,0.0,0.0,0.0,4.0/
      DATA Q/1.0,0.0,1.0,0.0,0.0,1.0/
      DATA R/1.0,0.0,1.0/
C
      N=3
      M=2
      NU=6
      MU=3
      L=0
      MAX=100
      EPS=1.E-8
      IPT=1
      INP=1
      CALL MRICKL(A,B,Q,R,SL,P,PP,Z,AA,RINV,L,N,M,NU,MU,INP,IPT,EPS,
     +                                            MAX,KR,H)
      CALL NEWSL(RINV,B,P,Z,SL,N,M)
      WRITE(6,10)
      CALL MPRINT(SL,M,N)
C
   10 FORMAT(//,2X,*OPTIMUM SL*,/,2X,*----------*)
      STOP
      END
```

INPUT DATA

THE A MATRIX

	1	2	3
1	-.60000E+00	0.	.45000E+00
2	0.	-.40000E+01	.80000E+00
3	.40000E+01	-.16000E+01	-.80000E+00

THE B MATRIX

	1	2
1	-.45000E+00	0.
2	0.	0.
3	0.	.40000E+01

THE Q MATRIX

	1	2	3
1	.10000E+01		
2	0.	.10000E+01	
3	0.	0.	.10000E+01

THE R MATRIX

	1	2
1	.10000E+01	
2	0.	.10000E+01

INITIAL VALUE OF THE MATRIX SL = ZERO MATRIX

THE INVERSE OF R IS

	1	2
1	.10000E+01	
2	0.	.10000E+01

ITER.NO.	P		
1	-.44449E+01		
	.26666E+00	.12901E+00	
	-.79174E+00	-.10036E-01	.16961E+00
2	-.40791E+01		
	.11503E+00	.13031E+00	
	-.37744E+00	-.13504E-01	.15722E+00
3	-.36027E+01		
	.61080E-01	.13175E+00	
	-.25003E+00	-.18525E-01	.17063E+00
4	-.34317E+01		
	.58805E-01	.13167E+00	
	-.23370E+00	-.18630E-01	.17205E+00
5	-.34264E+01		
	.58567E-01	.13168E+00	
	-.23311E+00	-.18655E-01	.17211E+00
6	-.34263E+01		
	.58567E-01	.13168E+00	
	-.23311E+00	-.18655E-01	.17211E+00
7	-.34263E+01		
	.58567E-01	.13168E+00	
	-.23311E+00	-.18655E-01	.17211E+00

FINAL SOLUTION OF THE MATRIX RICCATI EQUATION

	1	2	3
1	-.34263E+01		
2	.58567E-01	.13168E+00	
3	-.23311E+00	-.18655E-01	.17211E+00

OPTIMUM SL

	1	2	3
1	.15419E+01	-.26355E-01	.10490E+00
2	-.93243E+00	-.74621E-01	.68845E+00

Example 2

Consider the system given by

$$\dot{\underset{\sim}{x}} = \begin{bmatrix} 0 & 0 \\ 1 & -\sqrt{2} \end{bmatrix} \underset{\sim}{x} + \begin{bmatrix} 1 \\ 0 \end{bmatrix} u .$$

The system is completely controllable, since the controllability matrix is nonsingular.

$$\underset{\sim}{U} = [\underset{\sim}{B} \quad \underset{\sim}{A}\underset{\sim}{B}] = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} , \quad \det[\underset{\sim}{U}] = 1 \neq 0 .$$

But, $\underset{\sim}{A}$ has the following eigenvalues

$$\lambda_1 = 0 , \ \lambda_2 = -\sqrt{2} .$$

An initial value of SL_0 is given, for example, by

$$\underset{\sim}{SL}_0 = [2 \quad -1] .$$

Thus,

$$\underset{\sim}{A}_0 = \underset{\sim}{A} - \underset{\sim}{B}*\underset{\sim}{SL} = \begin{bmatrix} 0 & 0 \\ 1 & -\sqrt{2} \end{bmatrix} - \begin{bmatrix} 1 \\ 0 \end{bmatrix} [2 \ -1] ,$$

$$= \begin{bmatrix} -2 & -1 \\ 1 & -\sqrt{2} \end{bmatrix} ,$$

which has $\lambda = -(1+ 1/\sqrt{2}) \pm j\sqrt{\sqrt{2}-(1/2)}$, i.e., with negative real part.

The exact solution is

$$\underset{\sim}{\overline{P}} = \begin{bmatrix} 2-\sqrt{2} & 3-2\sqrt{2} \\ 3-2\sqrt{2} & 6-4\sqrt{2} \end{bmatrix} = \begin{bmatrix} 0.585786 & 0.171573 \\ 0.171573 & 0.343146 \end{bmatrix} .$$

```
      DIMENSION A(2,2),B(2,1),Q(3),R(1),SL(1,2),P(3),PP(3),Z(1,2),
     +             AA(2,2),RINV(1),KR(3),H(3,3)
C
      DATA A/0.0,1.0,0.0,-1.414213562/
      DATA B/1.0,0.0/
      DATA Q/0.0,0.0,1.0/
      DATA R/1.0/
      DATA SL/2.0,-1.0/
C
      N=2
      M=1
      NU=3
      MU=1
      L=1
      MAX=100
      EPS=1.E-8
      IPT=1
      INP=1
      CALL MRICKL(A,B,Q,R,SL,P,PP,Z,AA,RINV,L,N,M,NU,MU,INP,IPT,EPS,
     +                                          MAX,KR,H)
      STOP
      END
```

INPUT DATA

THE A MATRIX

	1	2
1	0.	0.
2	.10000E+01	-.14142E+01

THE B MATRIX

	1
1	.10000E+01
2	0.

THE Q MATRIX

	1	2
1	0.	
2	0.	.10000E+01

THE R MATRIX

	1
1	.10000E+01

THE INITIAL VALUE OF THE SL MATRIX

	1	2
1	.20000E+01	-.10000E+01

THE INVERSE OF R IS

	1
1	.10000E+01

ITER.NO.	P	
1	.93365E+00	
	-.13270E+00	.61327E+00
2	.61787E+00	
	.14103E+00	.37301E+00
3	.58609E+00	
	.17125E+00	.34351E+00
4	.58579E+00	
	.17157E+00	.34315E+00
5	.58579E+00	
	.17157E+00	.34315E+00
6	.58579E+00	
	.17157E+00	.34315E+00

FINAL SOLUTION OF THE MATRIX RICCATI EQUATION

	1	2
1	.58579E+00	
2	.17157E+00	.34315E+00

Example 3

A system of order 9 was used as an example and the final result was obtained in only seven iterations. The problem is shown in the following listing.

```
 DIMENSION A(9,9),B(9,4),Q(45),R(10),SL(4,9),P(45),PP(45),Z(4,9),
*AA(9,9),RINV(10),KR(45),H(45,45)
 DATA A/-4.855,-0.072,0.449,0.029,2.824,-0.187,-3.509,0.058,2.375,
+        2.592,-5.678,2.208,.671,10.265,-7.362,10.644,1.342,8.057,
+        -2.444,1.522,-8.235,-.609,-21.147,3.956,-9.747,-1.217,
+                                                         -13.912,
+        -1.091,-0.418,-1.129,-3.233,-4.824,.514,-3.886,-2.466,
+                                                         -3.694,
+        2*0.0, 1.0, 6*0.0,
+        0.0, 1.0, 7*0.0,
+        1.0, 8*0.0,
+        3*0.0, 1.0, 5*0.0,
+        4*0.0, 1.0, 4*0.0 /
C
 DATA B/8*0.0, 1.0,
+        6*0.0, 1.0, 2*0.0,
+        7*0.0, 1.0, 0.0,
+        5*0.0, 1.0, 3*0.0/
 DATA R/1.,0.,1.,0.,0.,1.,0.,0.,0.,1./
C
 DATA Q/909.,126.,89.,588.,-38.,580.,204.,51.,94.,53.,-96.,1.,-86.,
*-17.,13.,-60.,-10.,-36.,-14.,6.,4.,-183.,-22.,-124.,-40.,20.,12.,
*37.,-60.,-10.,-36.,-14.,6.,4.,12.,4.,-90.,-15.,-54.,-21.,9.,6.,
*18.,6.,9./
C
 N=9
 M=4
 NU=45
 MU=10
 L=0
 EPS=1.E-4
 MAX=100
 IPT=1
 INP=1
 CALL MRICKL(A,B,Q,R,SL,P,PP,Z,AA,RINV,L,N,M,NU,MU,INP,IPT,EPS,
+                                             MAX,KR,H)
 STOP
 END
```

INPUT DATA

THE A MATRIX

	1	2	3	4	5	6	7	8	9
1	-.45550E+01	.25920E+01	-.24440E+01	-.13910E+01	0.	0.	.10000E+01	0.	0.
2	-.72000E-01	-.56780E+01	.15220E+01	-.41800E+00	0.	.10000E+01	0.	0.	0.
3	.44900E+00	.22080E+01	-.82350E+01	-.11290E+01	.10000E+01	0.	0.	0.	0.
4	.29000E-01	.67100E+00	-.60900E+00	-.32330E+01	0.	0.	0.	.10000E+01	0.
5	.28240E+01	.10265E+02	-.21147E+02	-.48240E+01	0.	0.	0.	0.	.10000E+01
6	-.16700E+00	-.73620E+01	.39560E+01	.51400E+00	0.	0.	0.	0.	0.
7	-.35090E+01	.10644E+02	-.97470E+01	-.38860E+01	0.	0.	0.	0.	0.
8	.56000E-01	.13420E+01	-.12170E+01	-.24660E+01	0.	0.	0.	0.	0.
9	.23750E+01	.80570E+01	-.13912E+02	-.36940E+01	0.	0.	0.	0.	0.

THE B MATRIX

	1	2	3	4
1	0.	0.	0.	0.
2	0.	0.	0.	0.
3	0.	0.	0.	0.
4	0.	0.	0.	0.
5	0.	0.	0.	0.
6	0.	0.	0.	.10000E+01
7	0.	.10000E+01	0.	0.
8	0.	0.	.10000E+01	0.
9	.10000E+01	0.	0.	0.

THE Q MATRIX

	1	2	3	4	5	6	7	8	9
1	.90900E+03								
2	.12600E+03	.69000E+02							
3	.56800E+03	-.38000E+02	.58000E+03						
4	.20400E+03	.51000E+02	.94000E+02	.53000E+02					
5	-.96000E+02	.10000E+01	-.86000E+02	-.17000E+02	.13000E+02				
6	-.60000E+02	-.10000E+02	-.36000E+02	-.14000E+02	.60000E+01	.40000E+01			
7	-.18300E+03	-.22000E+02	-.12400E+03	-.40000E+02	.20000E+02	.12000E+02	.37000E+02		
8	-.60000E+02	-.10000E+02	-.36000E+02	-.14000E+02	.60000E+01	.40000E+01	.12000E+02	.40000E+01	
9	-.90000E+02	-.15000E+02	-.54000E+02	-.21000E+02	.90000E+01	.60000E+01	.18000E+02	.60000E+01	.90000E+01

THE R MATRIX

	1	2	3	4
1	.10000E+01			
2	0.	.10000E+01		
3	0.	0.	.10000E+01	
4	0.	0.	0.	.10000E+01

INITIAL VALUE OF THE MATRIX SL = ZERO MATRIX

THE INVERSE OF R IS

	1	2	3	4
1	.10000E+01			
2	0.	.10000E+01		
3	0.	0.	.10000E+01	
4	0.	0.	0.	.10000E+01

ITER. NO.	P								
1	.13049E+03								
	.18957E+02	.94012E+01							
	.72268E+02	.56371E+01	-.67145E+02						
	.24529E+02	.73126E+01	.13990E+02	.61287E+01					
	-.62753E+01	.95010E+01	-.65[illegible]E+01	-.56954E+00	.46003E+01				
	-.90665E+01	-.20000E+01	-.69501E+01	-.19311E+01	-.39374E-01	.13830E+01			
	-.13500E+02	-.29335E+01	-.13725E+02	-.41576E+01	.17242E+01	.16442E+01	.35417E+01		
	-.78422E+01	-.26539E+01	-.543[illegible]5E+01	-.20000E+01	.13354E+00	.12639E+01	.14443E+01	.14434E+01	
	-.19724E+02	-.59506E+01	-.136[illegible]E+02	-.61335E+01	-.45000E+01	.25785E+01	.27556E+01	.18895E+01	.12261E+02
2	.72609E+02								
	.12482E+02	.75565E+01							
	.51049E+02	.69013E+00	.50594E+02						
	.17583E+02	.54940E+01	.86732E+01	.51633E+01					
	-.63478E+01	.43478E+00	-.61212E+01	-.94201E+00	.20817E+01				
	-.56856E+01	-.10454E+01	-.43703E+01	-.12091E+01	.10977E+00	.85619E+00			
	-.13734E+02	-.19138E+01	-.10068E+02	-.31132E+01	.15136E+01	.10899E+01	.26892E+01		
	-.50390E+01	-.12211E+01	-.33097E+01	-.11795E+01	.17011E+00	.75881E+00	.95839E+00	.85722E+00	
	-.11218E+02	-.32307E+01	-.75879E+01	-.34547E+01	-.20826E+01	.13382E+01	.16382E+01	.10024E+01	.62498E+01
3	.68366E+02								
	.11216E+02	.76432E+01							
	.47671E+02	-.34155E+00	.47739E+02						
	.15333E+02	.50473E+01	.76556E+01	.57371E+01					
	-.67539E+01	.43057E-01	-.62018E+01	-.11652E+01	.19085E+01				
	-.53780E+01	-.79073E+00	-.38613E+01	-.79017E+00	.30823E+00	.72536E+00			
	-.13110E+02	-.17532E+01	-.95933E+01	-.29514E+01	.14622E+01	.10217E+01	.25808E+01		
	-.45610E+01	-.10045E+01	-.23139E+01	-.99360E+00	.29580E+00	.64014E+00	.91210E+00	.72018E+00	
	-.92297E+01	-.22302E+01	-.61621E+01	-.26758E+01	-.60206E+00	.84637E+00	.15100E+01	.65507E+00	.35047E+01
4	.67626E+02								
	.11110E+02	.69796E+01							
	.46862E+02	-.44131E+00	.46753E+02						
	.16160E+02	.50029E+01	.74681E+01	.56824E+01					
	-.63169E+01	-.47604E-01	-.57547E+01	-.11444E+01	.12483E+01				
	-.50754E+01	-.76022E+00	-.33832E+01	-.77589E+00	.33839E+00	.70815E+00			
	-.12965E+02	-.17723E+01	-.94427E+01	-.23260E+01	.13254E+01	.10321E+01	.25436E+01		
	-.45928E+01	-.98817E+00	-.29731E+01	-.97801E+00	.37807E+00	.62683E+00	.92333E+00	.70077E+00	
	-.95101E+01	-.20539E+01	-.64921E+01	-.26525E+01	.13641E+00	.73064E+00	.16433E+01	.56204E+00	.25745E+01
5	.67378E+02								
	.11102E+02	.69750E+01							
	.46540E+02	-.47230E+00	.46302E+02						
	.16107E+02	.49956E+01	.74014E+01	.56699E+01					
	-.61096E+01	-.37934E-01	-.54894E+01	-.11129E+01	.10568E+01				
	-.50853E+01	-.75962E+00	-.38941E+01	-.77753E+00	.33896E+00	.70720E+00			
	-.12915E+02	-.17707E+01	-.93834E+01	-.28157E+01	.12817E+01	.10347E+01	.25328E+01		
	-.46215E+01	-.99137E+00	-.30136E+01	-.98293E+00	.43305E+00	.62576E+00	.92856E+00	.69475E+00	
	-.96685E+01	-.23543E+01	-.66769E+01	-.26831E+01	.20547E+00	.72006E+00	.16695E+01	.54605E+00	.24393E+01

6

.67367E+02								
.11101E+02	.69747E+01							
.46524E+02	-.47453E+00	.46278E+02						
.16155E+02	.49933E+01	.73981E+01	.56693E+01					
-.61011E+01	-.36955E-01	-.54771E+01	-.11011E+01	.10502E+01				
-.50855E+01	-.75964E+00	-.38944E+01	-.77759E+00	.39916E+00	.70719E+00			
-.12913E+02	-.17705E+01	-.93810E+01	-.28153E+01	.12803E+01	.10348E+01	.25325E+01		
-.46231E+01	-.99164E+00	-.30214E+01	-.98309E+00	.40429E+00	.62574E+00	.92876E+00	.69436E+00	
-.96738E+01	-.20546E+01	-.66839E+01	-.26843E+01	.28961E+00	.71991E+00	.16815E+01	.54549E+00	.24364E+01

7

.67367E+02								
.11101E+02	.69747E+01							
.46524E+02	-.47453E+00	.46278E+02						
.16155E+02	.49933E+01	.73981E+01	.56693E+01					
-.61011E+01	-.36955E-01	-.54771E+01	-.11011E+01	.10502E+01				
-.50855E+01	-.75964E+00	-.38944E+01	-.77759E+00	.39916E+00	.70719E+00			
-.12913E+02	-.17705E+01	-.93810E+01	-.28153E+01	.12803E+01	.10348E+01	.25325E+01		
-.46231E+01	-.99164E+00	-.30214E+01	-.98309E+00	.40429E+00	.62574E+00	.92876E+00	.69436E+00	
-.96739E+01	-.20546E+01	-.66839E+01	-.26843E+01	.28961E+00	.71991E+00	.16815E+01	.54549E+00	.24364E+01

FINAL SOLUTION OF THE MATRIX RICCATI EQUATION

	1	2	3	4	5	6	7	8	9
1	.67367E+02								
2	.11101E+02	.69747E+01							
3	.46524E+02	-.47453E+00	.46278E+02						
4	.16155E+02	.49933E+01	.73981E+01	.56693E+01					
5	-.61011E+01	-.36955E-01	-.54771E+01	-.11011E+01	.10502E+01				
6	-.50855E+01	-.75964E+00	-.38944E+01	-.77759E+00	.39916E+00	.70719E+00			
7	-.12913E+02	-.17705E+01	-.93810E+01	-.28153E+01	.12803E+01	.10348E+01	.25325E+01		
8	-.46231E+01	-.99164E+00	-.30214E+01	-.98309E+00	.40429E+00	.62574E+00	.92876E+00	.69436E+00	
9	-.96738E+01	-.20546E+01	-.66839E+01	-.26843E+01	.28961E+00	.71991E+00	.16815E+01	.54549E+00	.24364E+01

PROGRAM LISTING

The Subroutine GELG*, called for solving the system of linear equations, is an SSP library subroutine.

---

* Subroutine GELG, System/360 Scientific Subroutine Package, Version III, IBM Programmer's Manual Number 360-CM-03x, p. 121.

```
      SUBROUTINE MRICKL(A,B,Q,R,SL,P,PP,Z,AA,RINV,L,N,M,NU,MU,INP,IPT,
C     ===============================================================
     *EPS,MAX,KR,H)
C
      DIMENSION A(N,N),B(N,M),Q(NU),R(MU),SL(M,N),P(NU),PP(NU),Z(M,N),
     *AA(N,N),RINV(MU),KR(NU),H(NU,NU)
C
C        WRITING OF INPUT DATA
      IF(INP.EQ.0) GO TO 10
      WRITE(6,250)
      CALL MPRINT(A,N,N)
      WRITE(6,260)
      CALL MPRINT(B,N,M)
      WRITE(6,270)
      CALL SMPRINT(Q,N)
      WRITE(6,280)
      CALL SMPRINT(R,M)
      IF(L.NE.0) GO TO 5
      WRITE(6,290)
      GO TO 10
    5 WRITE(6,295)
      CALL MPRINT(SL,M,N)
C
   10 CALL FACTOR(R,M,EPS,IER)
      IF(IER) 200,20,20
   20 DO 30 I=1,MU
      RINV(I)=R(I)
   30 CONTINUE
      CALL INVER(RINV,M,MU)
      WRITE(6,285)
      CALL SMPRINT(RINV,M)
      IR=0
      DO 32 I=1,N
      DO 32 J=I,N
      IR=IR+1
      JP=I+(J*J-J)/2
      KR(JP)=IR
   32 CONTINUE
      DO 35 I=1,NU
      KS=KR(I)
      I1=I+1
      DO 35 J=I1,NU
      IF(KR(J).EQ.I) KR(J)=KS
   35 CONTINUE
      IN=0
      WRITE(6,240)
      IF(L.NE.0) GO TO 50
      DO 40 I=1,NU
      P(I)=-Q(I)
   40 CONTINUE
      CALL SLEFORM(A,H,N,NU,IN,KR)
      GO TO 70
C
   50 CALL MSRBSL(R,SL,Z,N,M)
      CALL MTBSM(Z,P,M,N)
      DO 60 I=1,NU
      P(I)=-Q(I)-P(I)
   60 CONTINUE
      CALL AMBSL(A,B,SL,AA,N,M)
      CALL SLEFORM(AA,H,N,NU,IN,KR)
   70 CALL GELG(P,H,NU,1,EPS,IER)
      DO 75 I=1,NU
      KS=KR(I)
      IF(I.EQ.KS) GO TO 75
      CP=P(I)
      P(I)=P(KS)
      P(KS)=CP
   75 CONTINUE
      IN=IN+1
      IF(MOD(IN,IPT).EQ.0) CALL INTMPT(IN,P,N)
      IF(IER) 210,90,80
   80 WRITE(6,320)
   90 IF(IN.EQ.1) GO TO 110
      GE=0.0
```

```
      DO 100 I=1,NU
      E=ABS(P(I)-PP(I))
      IF(E.GT.GE) GE=E
  100 CONTINUE
C        STOPPING CRITERION
      IF(GE.LE.EPS) GO TO 220
C        CHECK FOR MAXIMUM ALLAWABLE NUMBER OF ITERATIONS
  110 IF(IN.GE.MAX) GO TO 230
      CALL NEWSL(RINV,B,P,Z,SL,N,M)
      DO 120 I=1,NU
      PP(I)=P(I)
  120 CONTINUE
      GO TO 50
C          END OF ITERATION LOOP
C
  200 WRITE(6,300)
      CALL EXIT
  210 WRITE(6,310)
      CALL EXIT
  220 WRITE(6,330)
      CALL SMPRINT(P,N)
      RETURN
  230 WRITE(6,340)
      CALL SMPRINT(P,N)
      RETURN
C
  240 FORMAT(1H1,2X,*ITER.NO.*,20X,*P*,/,2X,40(*-*))
  250 FORMAT(1H1,2X,*INPUT DATA*,/,3X,10(*-*),//,2X,*THE A MATRIX*,/)
  260 FORMAT(//,2X,*THE B MATRIX*,/)
  270 FORMAT(//,2X,*THE Q MATRIX*,/)
  280 FORMAT(//,2X,*THE R MATRIX*,/)
  285 FORMAT(//,2X,*THE INVERSE OF R IS*,/)
  290 FORMAT(//,4X,*INITIAL VALUE OF THE MATRIX SL = ZERO MATRIX*,/)
  295 FORMAT(//,2X,*THE INITIAL VALUE OF THE SL MATRIX*,/)
  300 FORMAT(//,4X,*R IS NOT POSITIVE DEFINITE*)
  310 FORMAT(//,4X,*SINGULAR MATRIX FOR THE SYSTEM OF LINEAR EQUATIONS*)
  320 FORMAT(//,4X,*LOST SIGNIFICANCE IN SOLVING THE SYSTEM OF LINEAR
     -EQUATIONS*)
  330 FORMAT(1H1,4X,*FINAL SOLUTION OF THE MATRIX RICCATI EQUATION*,/,
     -5X,45(*-*))
  340 FORMAT(1H1,4X,*MAXIMUM NUMBER OF ITERATIONS HAS BEEN ACHIEVED*,/,
     -4X,*FINAL VALUE OF P IS*)
      END
```

```
      SUBROUTINE SLEFORM(A,H,N,NU,IN,KR)
C     ==================================
C
C       THIS SUBROUTINE FORMULATES THE COEFFICIENT MATRIX H
C          OF THE SYSTEM OF LINEAR EQUATIONS   H*P=-Q-SLT*R*SL
C
      DIMENSION A(N,N),H(NU,NU),KR(NU)
C
C      INITIALIZATION OF H
      DO 10 I=1,NU
      DO 10 J=1,NU
      H(I,J)=0.0
   10 CONTINUE
C
C         FORMULATION OF THE MATRIX OF THE SYSTEM OF SIMULTANEUS
C         LINEAR EQUATIONS
   15 IE=1
      DO 50 K=1,N
      IS=IE-K
      DO 20 I=K,N
      DO 20 J=K,N
      IH=I+IS
      JH=J+IS
      C=A(J,I)
      IF(I.EQ.J) C=C+A(K,K)
      IF(I.EQ.K.AND.I.NE.J) C=C+C
      H(IH,JH)=C
   20 CONTINUE
      IE=IH+1
      IF (K.EQ.N) GO TO 50
      IC=IS+K+1
      IR=IH
      DO 40 J=IC,IH
      JL=J-IS
      IR=IR+1
      C=A(K,JL)
      H(IR,J)=C+C
      H(J,IR)=A(JL,K)
      IF (JL.EQ.N) GO TO 40
      IT=J
      J1=JL+1
      DO 30 L=J1,N
      IR=IR+1
      IT=IT+1
      H(IR,J)=A(K,L)
      H(J,IR)=A(L,K)
      H(IR,IT)=A(K,JL)
      H(IT,IR)=A(JL,K)
   30 CONTINUE
   40 CONTINUE
   50 CONTINUE
      DO 80 I=1,NU
      KS=KR(I)
      IF(I.EQ.KS) GO TO 80
      DO 70 J=1,NU
      C=H(I,J)
      H(I,J)=H(KS,J)
      H(KS,J)=C
   70 CONTINUE
   80 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE FACTOR(R,M,EPS,IER)
C     ==============================
C
      DIMENSION R(1)
C
      DOUBLE PRECISION DPIV,DSUM
C
C        TEST ON WRONG INPUT PARAMETER M
C
      IF(M-1) 12,1,1
    1 IER=0
C
C            INITIALIZE DIAGONAL-LOOP
C
      KPIV=0
      DO 11 K=1,M
      KPIV=KPIV+K
      IND=KPIV
      LEND=K-1
C
C             CALCULATE TOLERANCE
C
      TOL=ABS(EPS*R(KPIV))
C
C           START FACTORIZATION LOOP OVER K-TH ROW
C
      DO 11 I=K,M
      DSUM=0.D0
      IF(LEND) 2,4,2
C
C           START INNER LOOP
C
    2 DO 3 L=1,LEND
      LANF=KPIV-L
      LIND=IND-L
    3 DSUM=DSUM+DBLE(R(LANF)*R(LIND))
C
C           END OF INNER LOOP
C
C          TRANSFORM ELEMENT R(IND)
C
    4 DSUM=DBLE(R(IND))-DSUM
      IF(I-K) 10,5,10
C
C          TEST FOR NEGATIVE PIVOT ELEMENT AND LOSS OF SIGNIFICANCE
C
    5 IF(SNGL(DSUM)-TOL) 6,6,9
    6 IF(DSUM) 12,12,7
    7 IF(IER) 8,8,9
    8 IER=K-1
C
C            COMPUTE PIVOT ELEMENT
C
    9 DPIV=DSQRT(DSUM)
      R(KPIV)=DPIV
      DPIV=1.D0/DPIV
      GO TO 11
C
C          CALCULATE TERMS IN ROW
C
   10 R(IND)=DSUM*DPIV
   11 IND=IND+I
C
C          END OF DIAGONAL LOOP
C
      RETURN
   12 IER=-1
      RETURN
      END
```

```
      SUBROUTINE INVER(R,M,MU)
C     ========================
C
      DIMENSION R(1)
C
      DOUBLE PRECISION DIN,WORK
C
C        INVERT UPPER TRIANGULAR MATRIX
C
      IPIV=MU
      IND=MU
C
C          INITIALIZE INVERSION LOOP
C
      DO 5 I=1,M
      DIN=1.D0/DBLE(R(IPIV))
      R(IPIV)=DIN
      MIN=M
      KEND=I-1
      LANF=M-KEND
      IF(KEND) 4,4,1
    1 J=IND
C
C          INITIALIZE ROW-LOOP
C
      DO 3 K=1,KEND
      WORK=0.D0
      MIN=MIN-1
      LHOR=IPIV
      LVER=J
C
C          START INNER LOOP
C
      DO 2 L=LANF,MIN
      LVER=LVER+1
      LHOR=LHOR+L
    2 WORK=WORK+DBLE(R(LVER)*R(LHOR))
C
C          END OF INNER LOOP
C
      R(J)=-WORK*DIN
    3 J=J-MIN
C          END OF ROW LOOP
C
    4 IPIV=IPIV-MIN
    5 IND=IND-1
C
C          END OF INVERSION LOOP
C
C             CALCULATE INVERSE R BY MEANS OF INVERSE T
C           INVERSE R=INVERSE T*TRANSPOSE(INVERSE T)
C
C            INITIALIZE MULTIPLICATION LOOP
C
      DO 7 I=1,M
      IPIV=IPIV+I
      J=IPIV
C
C        INITIALIZE ROW LOOP
C
      DO 7 K=I,M
      WORK=0.D0
      LHOR=J
C
C            START INNER LOOP
C
      DO 6 L=K,M
      LVER=LHOR+K-I
      WORK=WORK+DBLE(R(LHOR)*R(LVER))
    6 LHOR=LHOR+L
C
C            END OF INNER LOOP
C
      R(J)=WORK
```

```
    7 J=J+K
C
C        END OF ROW AND  MULTIPLICATION LOOP
C
    8 RETURN
      END


      SUBROUTINE LOCATE(I,J,IJ,N)
C     ===========================
C
C        THIS SUBROUTINE CALCULATES THE VECTOR SUBSCRIPT FOR THE ELEMENT
C                I,J OF A SYMMETRIC MATRIX (N*N)
C
      IF(I-J) 10,10,20
   10 IJ=I+(J*J-J)/2
      RETURN
   20 IJ=J+(I*I-I)/2
      RETURN
      END


      SUBROUTINE AMBSL(A,B,SL,AA,N,M)
C     ===============================
C
C        THIS SUBROUTINE CALCULATES    AA=A-B*SL
C
      DIMENSION A(N,N),B(N,M),SL(M,N),AA(N,N)
C
      DO 20 I=1,N
      DO 20 J=1,N
      SUM=0.0
      DO 10 K=1,M
      SUM=SUM+B(I,K)*SL(K,J)
   10 CONTINUE
      AA(I,J)=A(I,J)-SUM
   20 CONTINUE
      RETURN
      END


      SUBROUTINE MSRBSL(R,SL,Z,N,M)
C     =============================
C
C        THIS SUBROUTINE CALCULATES    Z=( RU OF FACTORIZED R)*SL
C
      DIMENSION R(1),SL(M,N),Z(M,N)
C
      DO 20 I=1,M
      DO 20 J=1,N
      SUM=0.0
      DO 10 K=1,M
      CALL LOCATE(I,K,IK,M)
      SUM=SUM+R(IK)*SL(K,J)
   10 CONTINUE
      Z(I,J)=SUM
   20 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE MTBSM(Z,P,M,N)
C     =========================
C
C       THIS SUBROUTINE   CALCULATES   P=ZT*Z  ,  Z=RU*SL
C
      DIMENSION  Z(M,N),P(1)
C
      IR=0
      DO 20 J=1,N
      DO 20 I=1,J
      IR=IR+1
      SUM=0.0
      DO 10 K=1,M
      SUM=SUM+Z(K,I)*Z(K,J)
   10 CONTINUE
      P(IR)=SUM
   20 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE NEWSL(RINV,B,P,Z,SL,N,M)
C     ===================================
C
C       THIS SUBROUTINE CALCULATES   SL=RINV*BT*P
C
      DIMENSION RINV(1),B(N,M),P(1),SL(M,N),Z(M,N)
C
      DO 20 I=1,M
      DO20 J=1,N
      SUM=0.0
      DO 10 K=1,N
      CALL LOCATE(K,J,KJ,N)
      SUM=SUM+B(K,I)*P(KJ)
   10 CONTINUE
      Z(I,J)=SUM
   20 CONTINUE
      DO 40 I=1,M
      DO 40 J=1,N
      SUM=0.0
      DO 30 K=1,M
      CALL LOCATE(I,K,IK,M)
      SUM=SUM+RINV(IK)*Z(K,J)
   30 CONTINUE
      SL(I,J)=SUM
   40 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE MPRINT(A,N,M)
C     ======================
C
C       THIS SUBROUTINE  PRINTS AN NXM MATRIX
C
      DIMENSION A(N,M)
C
      WRITE(6,30)  (I,I=1,M)
      DO 10  I=1,N
      WRITE(6,20)  I,(A(I,J),J=1,M)
   10 CONTINUE
   20 FORMAT(/,2X,I2,4X,10(E11.5,1X))
   30 FORMAT(//,10X,10(I2,10X))
      RETURN
      END
```

```
      SUBROUTINE SMPRINT(A,N)
C     ======================
C
C       THIS SUBROUTINE PRINTS THE LOWER TRIANGULAR PART OF A SYMMETRIC
C       MATRIX
C
      DIMENSION A(1)
C
      WRITE(6,30) (I,I=1,N)
      JF=0
      DO 10 I=1,N
      JS=JF+1
      JF=JF+I
      WRITE(6,20) I,(A(J),J=JS,JF)
   10 CONTINUE
   20 FORMAT(/,2X,I3,4X,10(E11.5,1X))
   30 FORMAT(//,10X,10(I3,10X))
      RETURN
      END
```

```
      SUBROUTINE INTMPT(IN,A,N)
C     =========================
C
C       THIS SUBROUTINE IS FOR PRINTING INTERMEDIATE VALUES OF P
C
      DIMENSION A(1)
C
      JF=1
      WRITE(6,20) IN,A(1)
      IF(N.EQ.1) RETURN
      DO 10 I=2,N
      JS=JF+1
      JF=JF+I
      WRITE(6,30) (A(J),J=JS,JF)
   10 CONTINUE
   20 FORMAT(//,2X,I3,4X,E11.5)
   30 FORMAT(9X,10(E11.5,1X))
      RETURN
      END
```

SOC-197

STEADY STATE SOLUTION OF THE MATRIX RICCATI EQUATION

H.L. Abdel-Malek

April 1978, No. of Pages: 34

Revised:

Key Words: Linear optimal control, linear regulators, quadratic matrix equations

Abstract: A brief review of methods proposed for solving the matrix Riccati equation is given. A FORTRAN listing of a computer program based on Kleinman iterative technique is included. It is shown how to formulate the system of simultaneous linear equations to be solved in the Kleinman method. Some illustrative examples are also given.

Description: Contains Fortran listing, user's manual. The listing contains 474 statements of which 137 are comment cards.

Related Work: SOC-121.

Price: $30.00.